




CREATE

A WORKING ECOSYSTEM!

with



lynxcoding.club



With funding from

Canada

TakingITGlobal
INSPIRE INFORM INVOLVE
INSPIRER INFORMER ENGAGER

DESCRIPTION

Creating a Working Ecosystem

Students (grades 3-10) will create a working ecosystem, including a three-level food chain.

In a lake, you will have algae that grows, small fish that eat algae, and a bigger fish that feeds on small fish.

Students will code this app using Lynx at **lynxcoding.club**

Important: *This is not a project for students new to Lynx.*

Students should have tried Activity Cards like Geometric Fun and Make an Interactive app.

Students will code the computer to:

- Create a background that includes a lake.
- Create an “algae” turtle that stamps itself to represent growth.
- Create a “prey” turtle that wanders around and eats (erases) algae when it finds any.
- Create a “predator” turtle that wanders around and eats the smaller fish.
- Algae and fish all use color detection to “stay” in the lake.
- Predator uses collision detection to detect small fish and eat them.

Success Criteria

Co-construct success criteria with your students.




Get a Lynx Account and Get Started

Get a LYNX Account, understand the layout, plan your project



Get a Lynx Account

Details at lynxcoding.club



We suggest:

- teacher gets a **School Administrator Account**
- students get **permanent Individual Accounts**
- teacher creates a '**club**' and invites all students

NO Account

You can try Lynx for free without an account, by clicking on **Create a Lynx Project** on the home page at lynxcoding.club.

FREE TRIAL Account

For full access, register (click **Login/Register** located at the top, right side of the Lynx web page).

INDIVIDUAL Account

Convert your trial account to a permanent individual account before end of trial period.

SCHOOL ADMINISTRATOR Account

Convert your trial account to a School Administrator account before end of trial period.

Accounts are free for Canadians thanks to a subsidy by the Government of Canada.



Layout

Share

Save

Add Objects

Files

Procedures

Clipart

Project Tree

Settings

My Projects

Commands

Help

My project - page1

Procedures

```
1 ; This is an example of a procedure. Type the  
   word DrawSquare in the Command Centre (the  
   area below the white Work Area)  
2  
3 to DrawSquare  
4 pendown  
5 repeat 4 [  
6   forward 100  
7   right 90  
8 ]  
9 end  
10
```

Procedure
Pane

Clipart
Pane

Name your
Project

Turtle

Work Area
(Page)

Command
Centre

HELP!

STEPS FOR STUDENTS

The big picture. There are seven big steps in this project.

1. Log in, look for the **All Projects** tab, open the **Template** folder and look for the **Ecosystem** project. Select it and click on **Edit** so you can create your own ecosystem.

2. Create a “lake” background.

3. Create an algae-turtle that moves around and grows.

4. Create a small algae-eating fish (the prey).

5. Create a bigger fish (predator) that feeds on small fish.

6. Create a counter for how much the preys eat, make the population grow when they eat enough.

7. Create a counter for how much the predators eat, make the population grow when they eat enough.





Open the Ecosystem Template

Log in & click on the All Projects Tab.
Look for the Templates folder and open Ecosystem.
Click on Edit and Ecosystem opens.
RENAME IT and SAVE IT AS YOUR OWN PROJECT.
Look at the clipart included.




Create the environment

Create a lake background and freeze it


1. Log in, head to **All Projects** and into the **Templates** folder, open the project **Ecosystem**. It contains clipart that you will find useful.
2. Give a meaningful name to your project, and save it. 
3. Click on this icon  to look at the clipart provided with this template. The first spot contains a BIG clipart that you can use as a background: a lake surrounded by land. Note the number of this clipart (**1**).
4. If there isn't already a turtle on the page, create a new turtle.
5. Give it the shape of the big lake clipart. Type this in the Command Centre: **setshape 1**
6. Stamp this turtle. Type this in the Command Centre: **stamp**

It is as if nothing happened, but what you are looking at now is a BIG turtle sitting over a BIG image of itself (the background). Type **setshape 0** in the Command Centre to return the turtle to its original shape.

To rename your project, click here...
And type a new name...



7. A lot is going to happen on this background. Freeze it so you can always return to this clean empty lake. Type this in the Command Centre: **freezebg**
Later, when you will use **cg** (clear **g**raphics) or **clean** to get rid of all the stamped algae, this frozen background will remain untouched.




Growing Algae

Create an “algae-turtle” that moves around while stamping itself, while staying in the lake.



Growing Algae (1)

Move, Stay in the Lake, Stamp



1. You still have a turtle on the page right? If not, choose **Turtle** in the + menu. Right-click on the turtle to open its dialog box and name it **Algae**. The name will become important later. Click **Apply**.

2. Look at the clipart again. The algae clipart is the green thing in spot 2. Type this in the Command Centre: **setshape 2**

3. Create a procedure that makes it move around. Click here to return to the Procedures Pane and write this:



to AlgaeMove
forever [forward 5 right random 360]
end


4. Type **AlgaeMove** in the Command Centre to try it. Let it run! See? The algae-turtle can leave the lake! Let's fix that.

First, click on stopall (■) to stop the **forever** process. You can use colour detection to make the algae-turtle stay in the lake. As it moves, the turtle can check the colour it is landing on.

- If it is **blue**, stamp yourself.
- It is **brown**, point towards the centre of the lake and move in that direction.

Colour detection

5. Right-click on the algae-turtle to open its dialog box again.



6. Then, choose your procedure **AlgaeMove** in the **On click** menu.

7. Finally, choose **New** in the **On colour** menu. Click **Apply**.

This creates a procedures such as this one in the Procedures Pane.

Read the gray comments

then delete them. The

Algae-turtle will run this


procedure each time it moves.

See next card.

```
5 to Algae_oncolour :prevColour :newColour
6 ; Use an instruction like this to do colour
7 ; Moving from any colour to red, even red t
8 ; Pick your own colour name and instruction
9 ; IF :NEWCOLOUR = "RED [BACK 10 RIGHT 180]
10 end
```

Growing Algae (2)

Move, Stay in the Lake, Stamp



Colour detection (continued)

Colours have numbers and names. For colour detection, the name **blue** corresponds to many shades of **blue**, same for **brown**. Place the algae-turtle in the lake and type this in the Command Centre:

```
show namefromcolour colorunder
```

```
blue
```

This is what Lynx replies.

1. Edit the colour detection procedure so it looks like this:

```
to Algae_Oncolour :prevColour :newColour
if :newColour = 'blue' [stamp]
if :newColour = 'brown' [bounce]
end
```

Bounce is a procedure that is not defined yet. You will do that next.

Click on the algae-turtle to run **AlgaeMove** again. As long as the algae-turtle is in the lake, it will move and stamp itself. If it hits the shore, you will get an error message: **I don't know how to bounce**.

This is OK for now.

Bouncing on the shore

There are many ways to bounce. The “trick” we are going to use here is to place an **invisible turtle** in the centre of the lake. When the algae or fish-turtle hits the shore, they will “point towards” the centre-turtle and move a bit.

2. Create a new turtle. It is in the centre of the page (lake). Open its dialog box and name it **centre**.

Name

centre

3. Create this procedure in the Procedures Pane:

```
to bounce
towards 'centre' forward 20
end
```

Now if you run the algae-turtle again, it should stay in the lake. **BUT IT IS GROWING WAY TOO FAST!**
You will fix that next.

4. You can hide the centre-turtle. **Click on it to make it “the current turtle”** then type **ht** in the Commands Centre. It stands for **hide turtle**. In its dialog box, you can also uncheck **Visible**.


Growing Algae (3)

Controlling the speed of the growth


Save your work!
📁

The algae is growing too quickly! Use a slider to control the “wait” between each “**move and stamp**” of the algae.

1. Choose **Slider** in the + menu. A slider appears on the page.
2. Right-click on the slider and name it **AlgaeDelay**.
3. Set the min to 0, the max to 50, and the current value to 5.



You should have something like this:



The name of the slider reports its current value. If you type **show AlgaeDelay** in the Command Centre, Lynx will return the current value, in this example 5.

4. Do you see where you can add a **wait** instruction in your procedures? The instruction **wait AlgaeDelay / 10** means “wait whatever number is reported by the slider, divided by 10”.

Change this:

to **AlgaeMove**
forever [forward 5 right random 360]
end

To this:

to **AlgaeMove**
forever [wait AlgaeDelay / 10 forward 5 right random 360]
end

Clean the graphics on the page and click on the algae-turtle to launch it again. Now play with the slider to control the speed of the growth.

Algae-eating fish

Create a small fish that moves around and eats algae when it finds any, all while staying in the lake.

The small fish will later be the prey of a larger fish



Creating the prey (small fish) ⁽¹⁾

Moving around

Save your work!

The prey-turtle has to do three things:

- It has to move around,
- It has to react to **green** (eat algae)
- It has to react to **brown** (bounce)

To **eat algae**, the prey-turtle will briefly set its shape to a **blue** ball, stamp itself, and return to its fish shape. This “erases” algae.


To **bounce**, the prey-turtle can use the same **bounce** procedure as the algae.

Go to the Clipart pane and note the clipart number of the small fish (3). Also note the clipart number of the blue ball (4).

- Create a new turtle. Give it the shape of a small fish: type this in the Command Centre:
setshape 3
- The prey-turtle has its own way of moving. Write this new **PreyMove** procedure in the Procedures Pane:
to PreyMove
forever [forward 2 wait 1 rt -30 + random 61]
end


Random 61 will return numbers between 0 and 60. Therefore, the sum of -30 and that random number will be between -30 and +30. In other words, the prey-turtle will make a **right** or a **left** turn of 30 degrees (**right -30** is the same as **left 30**). Clever!

- Open the prey-turtle's dialog box, name this turtle **Prey**
- And set its **On Click** instruction to **PreyMove**.
- Click on the turtle to test it. It should move, but since it doesn't do colour detection, it goes anywhere on the page. Click on Stopall () to stop it.



Creating the prey (small fish) (2)

Colour detection (eat, bounce)

Save your work!


Remember the clipart number 4? When eating algae, the prey-turtle will stamp this clipart to “erase” the algae.

1. Open the prey-turtle’s dialog box again and choose **New** in the **On Colour** menu.

On colour

New...

This creates a procedures such as this one in the Procedures Pane:

```
18 ▾ to Prey_oncolour :prevColour :newColour
19   ; Use an instruction like this to do colour
20   ; Moving from any colour to red, even red
21   ; Pick your own colour name and instruction
22   ; IF :NEWCOLOUR = "RED [BACK 10 RIGHT 180]
23   end
```

2. Delete the gray comments and write this procedure instead:

```
to Prey_OnColour :prevColour :newColour
if :newColour = 'brown' [bounce]
if :newColour = 'green' [PreyEatAlgae]
end
```

The first colour detection (**brown**) is simple, it is the same as for the algae. The second colour detection (**green**) calls a procedure that does not exist yet. Let’s work on that.

In pseudo-code (plain English), eating the algae means:

- Set your shape to blue ball
- Stamp yourself
- Set your shape back to what you were (prey-turtle)

3. Turn that into real Lynx code:

```
to PreyEatAlgae
  setshape 'BlueBall' ;temporarily change shape
  stamp
  setshape 'PreyArt' ;get shape back to normal
end
```

What is this “**BlueBall**” and “**PreyArt**” thing?

Clipart has numbers (you have seen them), but they can also have names. Go to the Clipart Pane and right-click on the blue ball. You can use this name (**BlueBall** - one word, no space) with **setshape**.

Name

BlueBall

#4

4. Test the prey-turtle. Click on the algae-turtle and let it go. Click on the prey-turtle and see it “eat” algae when it finds some.



Predator fish

Create a larger fish (predator) that moves around and eats the small fish (prey).



Creating the predator (big fish) ⁽¹⁾

Moving around

The predator-turtle has to do two things:

- A. It has to move around,
- B. It has to react to **brown** (bounce)

It does not react to **green**, because it is carnivorous. It will eat the prey-turtle later in this project, using **collision detection**, not **colour detection**.

Go to the Clipart pane and note the clipart number of the big fish (5). You can also use its name: **PredatorArt1**.

1. Create a new turtle. Give it the shape of a big fish: type this in the Command Centre:
setshape 5
setsize 20 To make it smaller. Default size is 40.
2. The predator-turtle moves the same way as the prey-turtle, just a bit faster. You can always adjust its speed as you wish to increase or decrease its ability to catch preys - just change the **forward** or **wait** values. Write this new **PredatorMove** procedure in the Procedures Pane:

```
to PredatorMove  
  forever [forward 6 wait 1 rt -30 + random 61]  
end
```

3. Open the predator-turtle's dialog box, name this turtle **Predator**.
4. Set its **On Click** instruction to **PredatorMove**.
5. Click on the turtle to test it. It should move, but since it doesn't do colour detection yet, it goes anywhere on the page. Click on Stopall () to stop it.




Name	Predator	← 3
Xcor	-112	
On click	predatormove	← 4

Save your work!

Creating the predator (big fish) (2)

Colour detection (bounce)

The predator-turtle does colour detection too... but it only reacts to **brown**. It does not react to **green** (algae) because it is carnivorous and will feed on the prey-turtles later on, using **collision** detection, not **colour** detection.

4. Test the predator-turtle. For now, it should just wander around and bounce on the shore. Use the **Stopall** button () to stop it.

1. Open the predator-turtle's dialog box again and choose **New** in the **On Colour** menu.

On colour

New...

This creates a procedure such as this one in the Procedures Pane:

```
38 to Predator_oncolour :prevColour :newColour
39 ; Use an instruction like this to do colour
40 ; Moving from any colour to red, even red to
41 ; Pick your own colour name and instructions
42 ; IF :NEWCOLOUR = "RED [BACK 10 RIGHT 180]
43 end
..
```

2. Delete the gray comments and write this procedure instead:
`to Predator_OnColour :prevColour :newColour`
`if :newColour = 'brown' [bounce]`
`end`

The colour detection (**brown**) is the same as the algae and the prey-turtle.



Prey: Growing and Shrinking Population

Let the games begin!

When the prey eats enough algae, the prey population grows.



Prey Health (1)

Using a slider as a “health” meter

Save your work!
📁


A slider can be used as a **meter**. The name of a slider, preceded by “**set**”, is a command that “sets” the value of a slider. If you have a slider named **PreyHealth**, the instruction **SetPreyHealth 50** will set its value at **50**.

Each time a prey eats algae, the “**health**” of the small fish increases on the slider. When the slider reaches the top limit, a new “prey” appears.

There a quite a bit of coding here. Let’s start by creating the **Health** meter.

1. Choose **Slider** in the **+** menu.
2. Right-click on the slider to open its dialog box. Set its name to **PreyHealth**, min to **0**, max to **50**, and current value to **25**.
3. Place the slider somewhere on the land.
4. Test these instructions in the Command Centre, and watch the slider being updated:

```
setpreyhealth 30  
setpreyhealth preyhealth + 1  
setpreyhealth preyhealth - 1
```



Prey Health (2)

Eating Algae Increases the Health of the Population

Save your work!
🔄

There is already a procedure for eating algae. You just have to add to it.

- Each time a prey eats some algae, the “health” of the species increases on the slider.
- When it reaches the top limit, a new “prey” appears, and the Health slider is reset to 50 (middle point).


1. Change the procedure:

```
to PreyEatAlgae
  setshape 'BlueBall'           ;temporarily change shape
  stamp
  setshape 'PreyArt'             ;get shape back to normal
end

to:
  to PreyEatAlgae
    setshape 'BlueBall'           ;temporarily change shape
    stamp
    setshape 'PreyArt'             ;get shape back to normal
    If PreyHealth = 50 [ setPreyHealth 25
                        clone who ;if slider at max, clone prey
                        everyone [clickon]] ;and clickon everyone to trigger the clone
    setPreyHealth PreyHealth + 1   ;increase the slider
  end
```



On colour 'green',
eat algae...



and increase
PreyHealth slider.



If slider = 50...



Clone myself and
reset slider.

Prey Health (3)

Being bitten by the Predator Reduces the Health of the Prey Population


Save your work!
📁

This involves a **collision detection**, and it is better to put that code into the Predator's dialog box. If a Predator touches any other turtle:

- Check if this is a prey (Predator won't eat other predators or algae).
- If it's a prey, decrease the health of the Prey population and increase the health of the Predator population.
- If the health of the Predator reaches the top, a new Predator appears.

Let's start by creating the Health meter for the predators:

1. Choose **Slider** in the + menu.
2. Right-click on the slider to open its dialog box. Set its name to **PredatorHealth**, min to **0**, max to **10**, and current value to **5**.
3. Place the slider somewhere on the land.




Note: At this point and from now on, you should test your ecosystem regularly. Click on the prey, the predator and the algae to make them run. Click on **Stopall** to stop the simulation. Since you haven't frozen any turtle, you can move the prey into the algae to help it "improve health", or move it closer to a predator to cause it to "reduce health", or play with the sliders to make things happen more quickly.

Also: As you experiment with this ecosystem, you may see error messages once in a while, for example when two predators hit the same prey. Do not worry about such messages.

Prey Health (4)

Being bitten by the Predator Reduces the Health of the Prey Population



For the moment, let's just take care of "reducing the health of the prey population".

The predator-turtle has an **On Click** instruction and **On Colour** instruction, but it does not have an **On Touch** instruction yet:

1. Open the dialog box of the predator-turtle.
2. Choose **New** in the **On Touch** menu.
3. This creates a procedure such as this one in the Procedures Pane:

Name	Predator	
Xcor	218.6966	Ycor
On click	predatormove	
On touch	New...	
On message	-	
On colour	predator_oncolour	

3

```
57 to Predator_touch :touchedturtle
58 ; Use instructions like these to s
   another turtle.
59 ; The variable :TOUCHEDTURTLE cont
60 ; SAY "OUCH!
61 end
```

2

4. Read, then delete the gray comments, and type these instructions instead:

```
to Predator_touch :touchedturtle
if and ( 'PreyArt' = ask :touchedturtle [shape])
  ( not :touchedturtle = 'prey')
  [ BitePrey :touchedturtle ]
end
```

```
to BitePrey :touchedturtle
setPreyHealth PreyHealth - 5
if PreyHealth < 5 [setPreyHealth 25 remove :touchedturtle]
end
```



This says: if you (predator) touch a turtle, check its **shape number** and check its **name**. If the shape name is **PreyArt**, then it is a prey, not an algae, and not another predator. Also, you don't want to delete the "original" prey named "prey", you will see why later.

The **BitePrey** subprocedure says: decrease the health of the prey population, and if the health is getting low, reset the health meter and remove the touched turtle.

Predator: Growing and Shrinking Population

When the predators eats enough prey,
the predator population grows.



Predator Health (1)

Being bit by the Predator Reduces the Health of the Prey Population

Save your work!
🔄

There is already a procedure for when a predator bites a prey. It says that the health of the **prey** population **decreases** when a prey is bitten.

You just have to add to it:

- Each time a predator bites a prey, the “health” of the **predator increases** on the slider.
- Again, if the predator’s health reaches the top limit, the predator clones itself.


1. The **BitePrey** procedure looks like this:

```
to BitePrey :touchedturtle
  setPreyHealth PreyHealth - 5
  if PreyHealth < 5 [setPreyHealth 50 remove :touchedturtle]
end
```

Add these two lines:


```
to BitePrey :touchedturtle
  setPreyHealth PreyHealth - 5
  if PreyHealth < 5 [setPreyHealth 50 remove :touchedturtle]
  if PredatorHealth = 10 [setPredatorHealth 5 clone who everyone [clickon]]
  setPredatorHealth PredatorHealth + 1
```

2. end



On Touch Prey...

increase slider value.



If slider = 10...



Clone myself and
reset the slider.

Resetting the Population and Restarting the Environment

**To reset the simulation, you need to remove all the extra fish
and clean the lake.**

If only it was that easy in real life!



Resetting the Population

Save your work!
📄

The four original turtles that you have created are called **Prey**, **Algae**, **Predator**, and **Centre** (remember that turtle is hiding). All the other turtles (those which have been cloned), got default names: **t4**, **t5**, **t6** and so on.


To reset the environment to its starting point, all you have to do is “remove” all the turtles whose name starts with a “t”.

The **superprocedure** **ResetPopulation** will tell every turtle to run the **subprocedure** **RemoveAllTs**.


The subprocedure **RemoveAllTs** says: if “t” is the first letter of my name (who means “myself”), then I am a clone, you can remove “myself”.

1. Create these procedures in the Procedures Pane, and create a button to execute the procedure **ResetPopulation**.
to ResetPopulation
 everyone [RemoveAllTs]
 end

 to RemoveAllTs
 if 't' = first who [remove who]
 end



Restarting the Environment



You should have one button that resets the populations and launches the experiment.

1. Create this procedures in the Procedures Pane:

to restart

clean

resetpopulation

setpreyhealth 25

setpredatorhealth 5

everyone [clickon]

end

; The background is frozen. Clean will just clean the algae from the previous experiment.

; This will remove all the cloned turtles, but not the original fish and algae.

; Start with an "average" health for the prey population.

; Same for the predators.

; The algae, the prey and the predator all have an onclick instruction. This will launch them.

2. Create a **button** (from the + menu). Right-click on it to open its Dialog Box. Label it as you wish and in **On Click** set its instruction to the **restart** procedure.

Name	button1
Label	RESTART
On click	restart



Challenges and Final Code



Challenges - Going Further

Go Fish!


The predator fish does not have a predator. Its population will always increase. YOU can be the predator: once in a while, click on a predator, to make it the current turtle, and click on the button **GO FISH!** When you click on a predator, it will stop moving. The “fishing” procedure will have the instruction **remove who**.

GO FISH!

```
71 to MeFishing
72 ; If the current turtle (who) is NOT a member of that list,
73 ; it can be removed
74 if not member? who [predator prey algae] [remove who]
75 end
```

Animated predator

Did you notice that there are two predator cliparts! When setting the shape of the predator turtle, use the instruction **setshape [5 6]**. The turtle will alternate between the two shapes each time it moves.




Directional Fish

You can use a paint program to create an “east” fish and a similar “west” fish. In the move procedure, each time the heading is set to a random number, you can check the current heading and set the “east” shape if the fish is heading east, and the “west” shape if the fish is heading west:

to PreyMove

```
forever [forward 2 wait 1 rt -30 + random 61
        ifelse heading > 180 [setshape ?West?] [setshape ?East?]]
end
```

Use your shape numbers instead of **?West?** and **?East?**



Make adjustments

Is the ecosystem working to your liking? There are plenty of things you can adjust:

- The limits of the sliders;
- The speed of the prey or predator;
- The size of the algae stamper;
- The gain or loss of health (the amount) when eating happens.

Study the procedures and figure out what parameters are under your control.

The Entire Final Code!

Cheat sheet to help you with debugging

Save your
work!



```
to AlgaeMove
  forever [wait AlgaeDelay / 10 forward 5 right random 360]
end
```

```
to Algae_Oncolour :prevColour :newColour
  if :newColour = 'blue' [stamp]
  if :newColour = 'brown' [bounce]
end
```

```
to bounce
  towards 'centre' forward 20
end
```

```
to PreyMove
  forever [forward 2 wait 1 rt -30 + random 61]
end
```

```
to Prey_OnColour :prevColour :newColour
  if :newColour = 'brown' [bounce]
  if :newColour = 'green' [PreyEatAlgae]
end
```

```
to PreyEatAlgae
  setshape 'BlueBall' ;temporarily change shape
  stamp
  setshape 'PreyArt' ;get shape back to normal
  if PreyHealth = 50 [setPreyHealth 25 clone who everyone [clickon]]
  setPreyHealth PreyHealth + 1
end
```

```
to PredatorMove
  forever [forward 6 wait 1 rt -30 + random 61]
end
```

```
to Predator_OnColour :prevColour :newColour
  if :newColour = 'brown' [bounce]
end
```

```
to Predator_touch :touchedturtle
  if and ('PreyArt' = ask :touchedturtle [shape])
    (not :touchedturtle = 'prey')
    [BitePrey :touchedturtle]
end
```

```
to BitePrey :touchedturtle
  setPreyHealth PreyHealth - 5
  if PreyHealth < 5 [setPreyHealth 25 remove :touchedturtle]
  if PredatorHealth = 10 [setPredatorHealth 1 clone who everyone [clickon]]
  setPredatorHealth PredatorHealth + 1
end
```

```
to restart
  clean
  resetpopulation
  setpreyhealth 25
  setpredatorhealth 5
  everyone [clickon]
end
```

```
to resetpopulation
  everyone [RemoveAllTs]
end
```

```
to RemoveAllTs
  ; removes t4, t5, t6 and so on
  ; does not remove prey, predator, algae, centre
  if 't' = first who [remove who]
end
```

```
to MeFishing
  ; You can't fish the original predator, whose name
  is 'predator'
  if not who = 'predator' [remove who]
end
```

Note: This cheat sheet by itself is not sufficient to make the project work. You need to create the turtles and the sliders as described in this project for these procedures to work.

CODEtoLEARN

Credits

Principal Writer **Alain Tougas**
Contributors **Elena Yakovleva**
Michael Quinn
Sergei Soprunov



Create a Working Ecosystem by [Code To Learn](#) is licensed
under a [Creative Commons](#)
[Attribution-NonCommercial-ShareAlike 4.0 International License](#).

Partners



A program of



Connected North

With funding from

